```
> with(OreModules):
> with(OreMorphisms):
> with(Stafford):
> with(linalg):
```

Let us consider the first Weyl algebra $A = A_1(\mathbb{Q})$, where $\mathbb{Q}$ is the field of rational numbers,

```
> A := DefineOreAlgebra(diff=[d,t], polynom=[t]):
```

and the left A-module M finitely presented by the matrix R defined by

```
> R := evalm([[d,0,-t]]);
```

$$R := \begin{bmatrix} d & 0 & -t \end{bmatrix}$$

and the left A-module M' finitely presented by the matrix R' defined by

```
> Rp := evalm([[d,-t,0,0,0,-1],[0,d,0,-t,0,0],[0,0,d,0,-t,0]]);
```

$$Rp := \begin{bmatrix} d & -t & 0 & 0 & 0 & -1 \\ 0 & d & 0 & -t & 0 & 0 \\ 0 & 0 & d & 0 & -t & 0 \end{bmatrix}$$

Let us also consider the matrix P defined by

```
> P := evalm([[0,0,1,0,0,0],[1,0,0,0,0,1],[0,0,0,0,1,0]]);
```

$$P := \begin{bmatrix} 0 & 0 & 1 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 1 & 0 \end{bmatrix}$$

and the matrix P' defined by

```
> Pp := evalm([[0,0,1]]);
```

$$Pp := \begin{bmatrix} 0 & 0 & 1 \end{bmatrix}$$

which are such that R P = P' R'. Thus, they define the left A-homomorphism ι from M to M' induced by P. We can check that ι is injective:

```
> TestInj(R,Rp,P,A);
```

$$true$$

Hence, we get that M is isomorphic to the left A-submodule ι(M) = ($A^{1\times 3}$ P + $A^{1\times 3}$ R') / ($A^{1\times 3}$ R') of M'.

Let us now compute an element m* of M such that ι(m*) is a unimodular element of M'.

```
> U := UnimodularElementInSubmodule(R,Rp,P,A):
```

The output U of the command *UnimodularElementInSubmodule* contains two entries

```
> nops(U);
```

$$2$$

the first one U[1], namely,

```
> U[1];
```

$$\begin{bmatrix} 0 & 1 & 0 \end{bmatrix}$$

represents an elements m* of M which is such that ι(m*) is a unimodular element of M'. The second entry U[2] of U, namely,

```
> map(collect,U[2],[d,t]);
```

$$\begin{bmatrix} -\frac{2}{9}\,t^2 + t - \frac{1}{3}\,d\,t^2 + \frac{2}{27}\,t^3 \\ -\frac{t\,d^2}{3} + (\frac{1}{3} - \frac{5}{9}\,t + \frac{2}{27}\,t^2)\,d + \frac{5}{9} + \frac{2\,t^2}{27} \\ 0 \\ \frac{4}{27} - \frac{d^3}{3} + (-\frac{5}{9} + \frac{2\,t}{27})\,d^2 + (\frac{2\,t}{27} + \frac{4}{27})\,d \\ 0 \\ \frac{1}{3}\,d\,t^2 - t + 1 + \frac{2}{9}\,t^2 - \frac{2}{27}\,t^3 \end{bmatrix}$$

defines a left A-homomorphism ϕ from M' to A which is such that $\phi(\iota(\text{m}^*))$ = U[1] P U[2] = 1. Indeed, if λ^* = U[1] P, i.e.,

```
> lambda_star := Mult(U[1],P,A);
```

$$lambda\_star := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

then λ^* U[2] = U[1] P U[2] is equal to:

```
> Mult(lambda_star,U[2],A);
```

$$\begin{bmatrix} 1 \end{bmatrix}$$

Finally, let us check that ϕ is a well-defined left A-homomorphism from M' to A, i.e., R' U[2] = 0:

```
> Mult(Rp,U[2],A);
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$